

Themenübersicht Oktober 2002

Ausgabe: 10 / 2002

- Nice to know
- Schnelle Spannungsauswertung nach einer PSD Analyse
- Beitrag zur nichtlinearen Beulanalyse
- Flying Command Prompt
- Beschleunigung auf MASS21 Elemente

- **Unter anderem in der nächsten Ausgabe:**

Neue Features in der Version 7.0
Multi-Point-Constraint Elemente zur Lasteinleitung
Kantenkontakt

In eigener Sache:

Die Zusendung dieser Informationen erfolgt ausschließlich auf Wunsch des Empfängers und kann jederzeit unter www.cadfem.de beendet werden.

Wenngleich die vorliegenden Informationen mit größter Sorgfalt erstellt worden sind, weisen wir darauf hin, dass die Verwendung dieser unter Ausschluss jeglicher Gewährleistung erfolgt.

Nice to know

Ausgabe: 10 / 2002

ANSYS

- Der AUX3 stellt einen Editor für Resultfiles dar. Im AUX3 kann nur über die Kommandozeile gearbeitet werden. Mit dem DELETE Befehl können Teile des Resultfiles als zu löschend gekennzeichnet werden. Das Schreiben des modifizierten Resultfiles erfolgt erst auf den Befehl COMPRESS hin.
- Im AUX3 können auch Ergebnisgrößen (z.B. Zeitpunkte) geändert werden. Dazu steht der Befehl MODIFY zur Verfügung. Als entsprechender ausführender Befehl ist hier LIST,2 einzugeben.
- Mit BCSOPT,,,,,-5 wird eine Statistik über die Speicheranforderung des Sparsesolvers ausgegeben.
- Durch Drücken auf die rechte Maustaste im Menübaum erhält man ein Kontextmenü, in dem man das Verhalten des Menüs steuern kann. Wenn der Haken bei UIDL Behaviour gesetzt ist werden geöffnete und nicht mehr benötigte Äste automatisch geschlossen.

Schnelle Spannungsauswertung nach einer PSD Analyse

Ausgabe: 10 / 2002

Problem:

In Abhängigkeit der konkreten Fragestellung ist es z.t. notwendig Spannungen nach einer PSD Rechnung auszuwerten. Sofern insbesondere Modes höherer Ordnung zu berücksichtigen sind, (die Moden aber vorab nicht bekannt sind) ist es i.a. notwendig alle Eigenmoden der PSD Berechnung zugrundezulegen. Bei größeren Modellen jedoch führt dies zu einem unnötigen Filehandling (infolge der großen RST - Files). Dieser Beitrag stellt eine Lösung vor, die nur jene Moden berücksichtigt, die für den Lastfall relevant sind.



Erläuterung:

Bei der Single Point Response Spektrum Analyse gibt es eine Möglichkeit, vor der Modalanalyse ein (Dummy)-Spektrum in eine bestimmte Raumrichtung zu definieren. Bei der anschließenden Expansion (MXPAND) kann ein Signifikanzlevel angegeben werden um nur jene Moden zu expandieren, die von diesem anregbar sind.

Mit den Ergebnissen der Modalanalyse kann man danach die PSD-Berechnung durchführen.

Beispiel:

Das nachfolgende Beispiel eines einseitig eingespannten Balkens demonstriert dieses Verfahren anhand eines (Dummy)-Spektrum in X Richtung. Dieses sorgt dafür, das nur die longitudinalen Eigenformen bei 1668, 5046 und 8649 Hertz expandiert werden.



Schnelle Spannungsauswertung nach einer PSD Analyse

Ausgabe: 10 / 2002

ANSYS Eingabesatz (ANSYS 6.1):

```
fini
/output,a
/cle
/output
/prep7

dum=1      ! 1= mit dummy spektrum
           ! 0= ohne dummy spektrum

et,1,3
r,1,,1,.,1,1
mp,ex,1,1e7
mp,nuxy,1,0.3
mp,dens,1,.,00025
n,1
n,11,30
fill
e,1,2
*repeat,10,1,1
d,1,all
fini
/solu
antype,modal
modop,lanb,10
*if,dum,eq,1,then
sed,1
svtype,3
freq,1,10000
sv,.,02,1,1
*endif
solve
fini
/solu
expass,on
mxpand,10,,,yes,1
solve
fini
/post1
set,list
```

Resultübersicht ohne Dummy-Spektrum:

SET	TIME/FREQ	LOAD STEP	SUBSTEP	CUMULATIVE
1	124.03	1	1	1
2	765.65	1	2	2
3	1668.4	1	3	3
4	2094.5	1	4	4
5	3975.9	1	5	5
6	5046.4	1	6	6
7	6329.1	1	7	7
8	8548.8	1	8	8
9	9068.8	1	9	9
10	12125.	1	10	10

Resultübersicht mit Dummy-Spektrum:

SET	TIME/FREQ	LOAD STEP	SUBSTEP	CUMULATIVE
1	1668.4	1	3	1
2	5046.4	1	6	2
3	8548.8	1	8	3

Beitrag zur nichtlinearen Beulanalyse

Ausgabe: 10 / 2002

Programm:

Eine Stabilitätsuntersuchung sollte bei schlanken Strukturen, die knick- oder beulgefährdet sind, stets durchgeführt werden. In ANSYS ist es möglich eine lineare und nichtlineare Stabilitätsanalyse durchzuführen, wobei die lineare Betrachtung lediglich die Verzweigungslast und die verformte Konfiguration liefert, in die das System bei weiterer Laststeigerung ausweichen könnte.

Das genauere Verfahren ist die nichtlineare Beulanalyse. Unter anderem ist es hier möglich Imperfektionen anzugeben. Hierfür sollen die Fragen erläutert wie prinzipiell die nichtlineare Beulanalyse durchgeführt wird, wie die Imperfektionen berücksichtigt werden und worauf bei der Angabe von Imperfektionen zu achten ist.

Erläuterung:

Die nichtlineare Beulanalyse liefert anders als die lineare Beulanalyse die vollständige Last Verformungskurve. Es ist möglich das Vor- und Nachbeulverhalten zu analysieren. Ferner können Imperfektionen berücksichtigt werden.

Als Imperfektion wird gerne eine Ausgangskonfiguration verwendet, welche affin zu der ersten Beulform einer linearen Stabilitätsuntersuchung gewählt wird.

Es ist jedoch zu beachten, dass das System- und Stabilitätsverhalten in der nichtlinearen Berechnung maßgeblich von der Art und Größe der Imperfektion abhängen kann.

Beispiel:



Betrachtet wird ein Viertel-Segment einer Kreiszyinderschalen unter Axialdruck. Aufgrund der Symmetrie wird nur ein Viertel des Systems modelliert. Symmetrierandbedingungen sind entsprechend zu setzen.

Dargestellt wird die Verschiebung senkrecht zur Schalenfläche gegenüber der gesamten Axialkraft. Damit diese einfach im /POST26 ermittelt werden kann, wird eine Knotenkopplung in Axialrichtung durchgeführt.

Die erste Beulform wird mit einem geeigneten Faktor derart multipliziert, dass eine Ausmittigkeit zum perfekten System senkrecht zur Schalenfläche entsteht. So wird die nichtlineare Analyse basierend auf der imperfekten Struktur durchgeführt.

Beachten Sie, dass abhängig von Größe und Richtung (Vorzeichen), also je nach Art der Imperfektionsmodellierung unterschiedliches Systemverhalten vorliegen kann, welches entweder in ein Stabilitäts- oder Spannungsproblem mündet. Die unten angegebenen Last-Verformungskurven ergeben sich für eine Imperfektion von 1/20 der Schalendicke mit unterschiedlichem Vorzeichen.

Die Entscheidung, welche Imperfektion nach Art und Größe der Realität am nächsten kommt ist vom Berechnungsingenieur zu treffen.



Beitrag zur nichtlinearen Beulanalyse

Ausgabe: 10 / 2002

ANSYS Eingabesatz (ANSYS 6.1):

```
fini
/clear
/prep7

kraft=1
pi=acos(-1)
radius=83.333333
winkel=5/radius/pi*180
dicke=0.1

et,1,181
mp,ex,1,3.4e7
mp,prxy,1,0.2
r,1,dicke
cyl4,0,0,radius,-winkel/2,0,winkel/2,5
vdele,all
csys,1

asel,s,loc,x,radius
asel,invert
adele,all,,1
allsel
esize,,20
amesh,all

nrotat,all

nselect,s,loc,z,0
sf,all,pres,1
nselect,s,loc,y,-winkel/2
d,all,ux,0
d,all,roty,0
nselect,s,loc,y,winkel/2
d,all,uy,0
d,all,rotx,0
d,all,rotz,0
nselect,s,loc,z,0
d,all,ux,0
nselect,s,loc,z,5
*get,master,node,,num,min
cp,1,uz,all
d,master,uz,0
d,all,rotx,0
d,all,roty,0
allsel
```

```
/solu
antype,static
pstres,on
solve
fini

/solu
antype,buckle
bucopt,lanb,2
mxpand,2

solve

/post1
set,first
*get,last,active,,set,time

nsort,u,x
*get,max_x_weg,sort,0,max
*get,knoten_max_x_weg,sort,0,imax

faktor=-(1/max_x_weg)*(dicke/20)

/prep7
upgeom,faktor,1,1,schale,rst
/solu
antype,static,new

kraft=1.25*last
nselect,s,loc,z,-0.1,0.1
sf,all,pres,kraft
allsel

nlgeom,on
autots,off
nsubst,100
outres,all,all
arclen,on,10,0.1
solve
/post26
nsol,2,knoten_max_x_weg,u,x
rforce,3,master,f,z
xvar,2
prod,4,3,,,,,-1,1,1
plvar,4
```

Flying Kommand-Prompt

Ausgabe: 10 / 2002

Problem:

Das mit der ANSYS Version 6 erstmals ausgelieferte neue GUI hat nunmehr alle bisher einzelnen Fenster in einen gemeinsamen Rahmen fest eingebunden. Wer nach wie vor mit einzelnen Fenstern arbeiten möchte findet in der beiliegenden Applikation einige Anregungen zum Umsetzung mit TCL.

Darüber hinaus erhält der Anwender bei Verwendung dieser Applikation die gewohnt funktionierende Kommando-Hilfe (ohne Zeitlimit) und den History- Buffer.

Installation:

Kern der Applikation bildet ein TCL Script (jnCMDL1.eui) welches in den ANSYS Installationspfad unter C:\Programme\Ansys Inc\ANSYS61\gui kopiert wird.

Anschließend muß das Script eingebunden werden. Dies geschieht indem der folgende Eintrag zur bestehender Datei tclIndex (unter C:\Programme\Ansys Inc\ANSYS61\gui) hinzugefügt wird:

```
set auto_index( :euidl : jncmdl : start ) [ list source [ file join $dir jncmdl1.eui ] ]
```

Zum flexibleren Start wird ein Makro mitgeliefert (jnCMDL.MAC) welches im DOC Verzeichnis abgelegt von der Kommandozeile startbar ist. Für etwas mehr Komfort sollte jedoch die start61.ans mit einer Abkürzung ergänzt werden, damit der Start auch wirklich per Knopfdruck gelingt.

Hierzu erweitern Sie bitte die start61.ans um folgenden Eintrag:


```
*abbr,KICK,~tcl,':euidl:jncmdl::start,
```

Handhabung:

Nach dem Start erhalten Sie eine frei schwebende Kommandozeile mit den wesentlichen Funktionen:

1. Kommando-Syntax
2. Direktzugriff zur Hilfe (Fragezeichen)
3. History Liste

Das Tools ist somit intuitiv bedienbar. Zur korrekten Dokumentation aber ist eine komplette Beschreibung der Handhabung im Header des TCL Scriptes beigelegt.



Das Script finden Sie wie immer auf unserer Homepage.

Beschleunigungsbelastung auf MASS21 Elemente

Ausgabe: 10 / 2002

Problem:

Das MASS 21 Element hat in Zusammenhang mit der Beschleunigungslast ACEL eine schlecht dokumentierte unerwartete Eigenschaft.

Der ACEL Befehl wirkt im Knotenkoordinatensystem. Erwartungsgemäß sollte er im globalen Koordinatensystem wirken.

Einen Hinweis zu dieser Eigenschaft findet man zwar im Element reference guide 2.3.2, nicht jedoch bei der command reference von ACEL.


Beispiel:

Das MASS21 Element kann neben Massen auch mit rotatorischen Hauptträgheitsmomenten versehen werden. Dies ist in der Modellierungspraxis als Ersatzmodell für nicht mitmodellerte Anbauten mit entsprechender Masse und räumlicher Erstreckung die Standardtechnik.

Zu diesem Zweck ist es jedoch erforderlich, das Elementkoordinatensystem mit der keyopt,,2,1 auf das entsprechend rotierte Knotenkoordinatensystem (Hauptachsensystem) zu setzen. Hat man ferner Beschleunigungslasten ACEL (quasistatische, harmonische oder transiente Analyse) im Modell, geht man laut ACEL command reference zunächst davon aus, daß diese ordentlich im globalen Koordinatensystem aufgebracht werden. Daß dem jedoch nicht so ist, zeigen die Reaktionskräfte des anhängenden Inputs. Vielmehr interpretiert ANSYS die Beschleunigung nun im Knotenkoordinatensystem ohne die korrekte Transformation vorzunehmen.

Abhilfe: Als Workaround wäre denkbar bei rotatorischen Trägheiten auf MASS21 vollkommen zu verzichten und stattdessen MATRIX27 Elemente zu verwenden. Dort existiert zwar das gleiche Problem, aber es ist möglich, das globale Koordinatensystem beizubehalten und die dann natürlich anfallenden deviatorischen Momente "zu Fuß" auszurechnen und einzutragen.

```
fini
/clear
/prep7
et,1,mass21
keyopt,1,1,0
keyopt,1,2,1
keyopt,1,3,4
r,1,1 ! mass
n,1
clocal,11,0,,,45
nrotat,1
e,1
fini
/solu
antype,static
d,all,all
acel,1
solve
/post1
prrsol
```



```
PRRSOL Command
File
PRINT REACTION SOLUTIONS PER NODE
**** POST1 TOTAL REACTION SOLUTION LISTING ****
LOAD STEP= 1 SUBSTEP= 1
TIME= 1.0000 LOAD CASE= 0
THE FOLLOWING X,Y,Z SOLUTIONS ARE IN GLOBAL COORDINATES
NODE FX FY
1 0.70711 0.70711
TOTAL VALUES
VALUE 0.70711 0.70711
```